


## Create a Column Chart with Relative Variance in Power BI

Using the native Line and Clustered Column chart



Created by: Michel van Schaik

LinkedIn: <https://www.linkedin.com/in/michel-van-schaik/>

## Contents

Introduction.....	3
Step 1 - Create a Supplemental Date Table.....	4
Step 2 - Create Base Measures .....	5
Step 3 - Create Visual-Specific Measures .....	5
Step 4 - Apply Dynamic Format Strings.....	10
Step 5 - Build the Visual.....	11
Step 6 - Visual Formatting.....	12
6a - Data Labels.....	16
6b - Series Labels .....	17
6c - Create Reference Lines.....	18
6d - Create Error Bars.....	21
Final Result.....	23
Appendix - DAX Measures Cheat Sheet .....	24

## Introduction

This guide explains how to build a **Column Chart with Relative Variance** in Power BI using the native **Line and Clustered Column** chart.

The visual combines:

- **Columns** for actual and comparison values
- **Error bars** for positive/negative relative variance, capped to prevent overlap
- **Data labels** showing both absolute and relative variance
- **Reference lines** for alignment

We'll use a **supplemental date table** to control Actuals display up to a selected month while showing full-year comparisons.

This approach is **entirely based on regular DAX**, because **Visual Calculations do not support data labels or error bars**, making them only very limitedly usable in this scenario.


To speed up the build, you can use the provided **TMDL script** to create all required DAX measures in one go, instead of adding them manually one by one. For a short explanation of how to do this, refer to the **Quick Guide** available in the same GitHub repository.

## Step 1 - Create a Supplemental Date Table

### Purpose:

The supplemental date table allows us to show Actuals **only until the selected month** while keeping full-year values for the Comparison columns.

1. Create a **duplicate** of your main Dim Calendar table (or generate a new one). Name it `dim_calendar_supplemental` for example.
2. Create an **inactive one-to-many relationship** between the Dim Calendar Supplemental table and Dim Calendar table.



3. In Dim Calendar Supplemental table, sort Month Name by Month Number to ensure proper axis order (sorting Month Name by Month Number avoids alphabetical order issues).

Want to learn more about the technique of using a supplemental calendar table? Check out this resources:

SQLBI

<https://www.sqlbi.com/articles/show-previous-6-months-of-data-from-single-slicer-selection/>

Goodly

<https://www.youtube.com/watch?v=k0I-khGGXE4>

## Step 2 – Create Base Measures

You need to create two base measures:

### 1. Actual value

```
_Revenue = SUM(fact_table[Revenue])
```

*(Replace Revenue with your metric, e.g., Gross Profit.)*

### 2. Comparison value

```
_Revenue PY =  
CALCULATE (  
    [_Revenue],  
    DATEADD ('dim_calendar'[Date], -1, YEAR )  
)
```

*(Replace with Budget Logic if needed)*

As used in  
sample  
.pbix

## Step 3 – Create Visual-Specific Measures

We'll create 15 visual-specific DAX measures to handle:

- Column values
- Variance & Data label calculations
- Error bars
- Data label positions
- Max Y-axis control
- Reference lines

You can:

- **Copy & paste** each measure manually
- Or run the **TMDL script** (available in the GitHub folder) to create them all at once.

**Important:** Replace the placeholders below with the corresponding names from your own model (Date, Month/Month Number, base measures):

**Placeholders that need to be replaced:**

```
dim_calendar[Date]  
dim_calendar[Month Number]  
'dim_calendar'[Date]  
'dim_calendar_supplemental'[Date]  
dim_calendar_supplemental[Month Number]  
'dim_calendar_supplemental'[Month Number]  
'dim_calendar_supplemental'[Month]  
[_Revenue]  
[_Revenue PY]
```

**Note:** Do not change the measure names at this stage – otherwise, it will be harder to follow the steps in this guide. Of course, you can rename the measures afterwards once everything is finished.

## DAX Measures (Placeholders that need to be replaced with your own model's names)

01. Month ≤ Selected Month =

// This measure checks whether the month from the supplemental date table is less than or equal to the month currently selected in the main date table

// Used as filter in measures to limit Data Labels so they are only shown up to the selected month

```
VAR __SelectedMonth = MAX ( dim_calendar[Month Number] )
VAR __SupplMonth = MAX ( dim_calendar_supplemental[Month Number] )

VAR __Result =
    IF (
        __SupplMonth <= __SelectedMonth,
        1,
        0
    )

RETURN __Result
```

02. Actual Value =

// show Actuals up to the selected period

```
VAR __StartDate =
    STARTOFYEAR ( dim_calendar[Date] )
VAR __EndDate =
    MAX ( dim_calendar[Date] )
VAR __DateRange =
    DATESBETWEEN ( dim_calendar[Date], __StartDate, __EndDate )
VAR __Result =
    CALCULATE (
        [ _Revenue ],
        __DateRange,
        USERRELATIONSHIP ( 'dim_calendar_supplemental'[Date], 'dim_calendar'[Date] )
    )
RETURN
    __Result
```

03. Comparison Value =

// Show Comparison value for all months

```
VAR __StartDate =
    STARTOFYEAR ( dim_calendar[Date] )
VAR __EndDate =
    ENDOFYEAR ( dim_calendar[Date] )
VAR __DateRange =
    DATESBETWEEN ( dim_calendar[Date], __StartDate, __EndDate )
VAR __Result =
    CALCULATE (
        [ _Revenue PY ],
        __DateRange,
        USERRELATIONSHIP ( 'dim_calendar_supplemental'[Date], 'dim_calendar'[Date] )
    )
RETURN
    __Result
```

```

04. Data label Position abs. var. =
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __Comparison = [03. Comparison Value]
VAR __Actual = [02. Actual Value]
VAR __MaxValue =
    MAX ( __Comparison, __Actual )
VAR __Result =
    IF ( __IsHistoricalMonth = 1, __MaxValue, BLANK () )
RETURN
    __Result

```

```

05. Data label Value abs. var. =
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __Actual = [02. Actual Value]
VAR __Comparison = [03. Comparison Value]
VAR __Variance = __Actual - __Comparison
VAR __Result =
    IF ( __IsHistoricalMonth = 1, __Variance, BLANK () )
RETURN
    __Result

```

```

06. Max Value of Columns =
VAR __MAX_Value_Actual =
    MAXX (
        ALLSELECTED ( 'dim_calendar_supplemental'[Month Number],
            'dim_calendar_supplemental'[Month] ),
        [02. Actual Value]
    )
VAR __MAX_Value_Comparison =
    MAXX (
        ALLSELECTED ( 'dim_calendar_supplemental'[Month Number],
            'dim_calendar_supplemental'[Month] ),
        [03. Comparison Value]
    )
VAR __Result = MAX(__MAX_Value_Actual, __MAX_Value_Comparison)
RETURN __Result

```

```

07. Ref. line Position pos. rel. var % =
[06. Max Value of Columns] * 1.95

```

```

08. Ref. line Position neg. rel. var % =
[06. Max Value of Columns] * 1.95

```

```

09. Data label Value pos. rel. var. % =
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __Actual = [02. Actual Value]
VAR __Comparison = [03. Comparison Value]
VAR __Variance = __Actual - __Comparison
VAR __RelativeVariance = DIVIDE ( __Variance, __Comparison )

VAR __Result =
  IF (
    __IsHistoricalMonth = 1 && __Variance >= 0,
    __RelativeVariance,
    BLANK()
  )

RETURN __Result

10. Data label Value neg. rel. var. % =
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __Actual = [02. Actual Value]
VAR __Comparison = [03. Comparison Value]
VAR __Variance = __Actual - __Comparison
VAR __RelativeVariance = DIVIDE ( __Variance, __Comparison )

VAR __Result =
  IF (
    __IsHistoricalMonth = 1 && __Variance < 0,
    __RelativeVariance,
    BLANK()
  )

RETURN __Result

11. Upper Bound error bar pos. rel. var. % =
// Scale (×1.5): making the bars longer and more prominent.
// Cap (50%): avoids overlaps when months have very large % deltas.
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __RelativeVariance = [09. Data label Value pos. rel. var. %]
VAR __MaxBarLength = [06. Max Value of Columns] * 1.5
VAR __MaxCap = 0.5
VAR __Result =
  IF(
    __IsHistoricalMonth = 1,
    MIN(__RelativeVariance, __MaxCap) * __MaxBarLength,
    BLANK()
  )
RETURN __Result

```



```

12. Lower Bound error bar neg. rel. var. % =
// Scale (x1.5): making the bars longer and more prominent.
// Cap (50%): avoids overlaps when months have very large % deltas.
VAR __IsHistoricalMonth = [01. Month ≤ Selected Month]
VAR __RelativeVariance = [10. Data label Value neg. rel. var. %]
VAR __MaxBarLength = [06. Max Value of Columns] * 1.5
VAR __MaxCap = -0.5
VAR __Result =
  IF(
    __IsHistoricalMonth = 1,
    MAX(__RelativeVariance, __MaxCap) * __MaxBarLength,
    BLANK()
  )
RETURN __Result

13. Data label Position pos. rel var. % =
IF (
  [11. Upper Bound error bar pos. rel. var. %] >= 0,
  [11. Upper Bound error bar pos. rel. var. %] + [07. Ref. line Position pos. rel. var %]
)

14. Data label Position neg. rel var. % =
IF (
  [12. Lower Bound error bar neg. rel. var. %] < 0,
  [12. Lower Bound error bar neg. rel. var. %] + [08. Ref. line Position neg. rel. var %]
)


15. Max Y-axis Value =
// This measure calculates the maximum Y-axis value needed for the visual and then
increases it slightly to guarantee enough space for data labels
MAXX (
  ALLSELECTED (
    'dim_calendar_supplemental'[Month Number],
    'dim_calendar_supplemental'[Month]
  ),
  [13. Data label Position pos. rel var. %]
) * 1.1

```

## Step 4 – Apply Dynamic Format Strings

### Required:

- Set dynamic format string for DAX measures **09** and **10** → display as percentages with plus sign (+) for positive percentages.




### Optional:

- Measures **02** and **05** → dynamic units according to preference.


If you use the **TMDL script**, format strings are already included for **09** and **10**.

## Step 5 - Build the Visual

- Add a **Year/Month slicer** from the main **Dim Calendar** (not supplemental).
- Insert a **Line and Clustered Column Chart**.




- Set **X-axis** to **Month Name** from the **Dim Calendar Supplemental** table.
- Assign columns and lines per the provided measure numbers (see screenshot below).



## Step 6 – Visual Formatting

In the visual formatting pane make sure:

- 'Title', 'Data labels' and 'Series Labels' are turned On
- 'Legend', 'Zoom slider' and 'Shade Area' are turned Off



## Title

Example: *Revenue Performance by Month*

Optional subtitle: *Actual vs Previous Year*

## X-Axis

- Values: On
- Title: Off

The screenshot shows the 'X-axis' configuration panel. It is divided into three main sections: 'Values', 'Title', and 'Layout'. The 'Values' section is expanded and contains settings for font (Calibri, size 9), bold/italic/underline options, color (black), maximum height (25%), and a 'Concatenate labels' toggle (Off). The 'Title' section has a 'Title' toggle (Off). The 'Layout' section has a 'Minimum category width' setting (20 px).

## Y-Axis

- Values & Title: Off
- Set measure [15. Max Y-axis Value] as Maximum Y-axis



The screenshot shows the 'Y-axis' configuration panel. It is divided into three main sections: 'Range', 'Values', and 'Title'. The 'Range' section is expanded and contains settings for 'Minimum' (Auto) and 'Maximum' (fx). The 'Maximum' setting is highlighted with a red box. Below the 'Range' section are 'Logarithmic scale' (Off) and 'Invert range' (Off) toggles. The 'Values' section has a 'Values' toggle (Off). The 'Title' section has a 'Title' toggle (Off).

### Secondary Y-Axis



- Align zeros: On
- Values & Title: Off

▼ Secondary y-axis

Minimum

Auto    $fx$

Maximum

Auto    $fx$

Logarithmic scale

☐ Off




Align zeros

☒ On

▼ Values



☐ Off

Font


Calibri  9  

**B** *I* U



Color

Display units

Auto 

Value decimal places


Auto  

> Title


☐ Off

## Columns

- For **Series** -> **All** simply copy the settings exactly as shown in the screenshot below




- Set Color for Series **03 - Comparison Value** (for example Grey #B3B3B3) & Series **02 - Actual Value** (for example Black #333333)



#### Lines & Markers:

- Off for all series



### 6a - Data Labels


- Use positions & value fields exactly as in the table below.
- For **02** & **04**, also set the **Display units** and **Decimal places** (or use dynamic formatting in the DAX measures instead).
- For measures **13** & **14**: **Display units** = **None**, **Decimal places** = **1**.

Series	Turn series	Options > Label Position	Value > Field
03. Comparison Value	Off	-	-
02. Actual Value	On	Inside Center	[02. Actual Value]
04. Data label Position abs. var.	On	Above	[05. Data label Value abs. var.]
07. Ref. line Position pos. rel. var %	Off	-	-
08. Ref. line Position neg. rel. var %	Off	-	-
13. Data label Position pos. rel var. %	On	Above	[09. Data label Value pos. rel. var. %]
14. Data label Position neg. rel var. %	On	Under	[10. Data label Value neg. rel. var. %]




## 6b - Series Labels


- Series: All → set Options as shown in the screenshot below, and make sure 'Show for this series' is turned **Off**.



- Turn **On** Series labels for **04. Data label Position abs. var.**



- Rename measure name **04. Data label Position abs. var.** in the Line y-axis to **ΔPY**.



## 6c - Create Reference Lines

### Purpose:

Reference lines serve as baseline for the error bars and as the zero line for the Y-axis.

#### 1. Add zero reference line

- Add a new line.
- Rename it to **AC**.
- Set Type = Y-Axis Constant Line.
- Set Line Value = 0
- Turn Data label On
- Turn Shade area Off
- Apply the remaining settings exactly as shown in the screenshots below.

Reference line (1)

Apply settings to

+ Add line

AC

Type

Y-Axis Constant Line

Line

Value

0

Color

Transparency

0 %

Line style

Solid

Width

2 px

Position

In front

Shade area

Off

Data label

On

Horizontal position

Left

Vertical position

Above

Style

Name

Color

Display units


Auto

Value decimal places

Auto

## 2. Add reference line error bars


- Add a new line.
- Rename it to  $\Delta PY\%$ .
- Set Type = Y-Axis Constant Line.
- Set Line Value = [07. Ref. line Position pos. rel. var %]
- Set color black
- Turn Data label On
- Turn Shade area Off
- Apply the remaining settings exactly as shown in the screenshots below.



### 3. Add duplicate (white) reference line error bars

- Add a new line.
- Rename it to **White line**.
- Set Type = Y-Axis Constant Line.
- Set Line Value = [07. Ref. line Position pos. rel. var %]
- Set color white
- Turn Data label Off
- Turn Shade area Off
- Apply the remaining settings exactly as shown in the screenshots below

**Note:** This reference line isn't strictly necessary, but the white line (reference line 3) with a black outline (reference line 2) gives a nice touch.





▼ Reference line (3)

Apply settings to

+ Add line

AC

White line  



ΔPY%

Type


Y-Axis Constant Line ▼

▼ Line


Value

Color

 ▼


Transparency

0 % 

Line style

Solid ▼

Width

2 px 

Position

In front ▼

## 6d – Create Error Bars

### Purpose:

Error bars visually represent the relative variance between Actual and Comparison values. They show how much the Actual value deviates (in %) from the Comparison value upward for positive variance and downward for negative variance.

In this setup, the bars are scaled ( $\times 1.5$ ) to make them more prominent and capped at  $\pm 50\%$  to prevent overlap with other chart elements.

The bar length serves as a visual indicator, while the data label displays the exact percentage.

### 1. Add Error Bar for Positive Relative Variance

- Select series 07. Ref. line Position pos. rel. var %.
- Go to Options → Enabled.
- Set Upper Bound = [11. Upper Bound error bar pos. rel. var. %].
- Relation to measure = **Relative** (this ensures the bar length is scaled correctly relative to the reference line position).
- Turn Bars and Markers **On**
- Turn Error Bands, Error Label and Tooltip **Off**.
- Apply the remaining settings exactly as shown in the screenshots below. (The width of the bar and the size of the markers may depend on your canvas size, so adjust as needed for the best appearance.)

The screenshot displays the configuration interface for error bars, divided into two main panels. The left panel, titled 'Error bars', contains the 'Apply settings to' section with 'Series' set to '07. Ref. line Position ...'. Below this is the 'Options' section, where 'Enabled' is turned on, 'Type' is 'By field', 'Upper bound' is set to '11. Upper B...', 'Lower bound' has a '+Add data' button, 'Relationship to measure' is 'Relative', and 'Make symmetrical' is turned off. The right panel shows 'Bar' settings with 'Bar' turned on, 'Match series color' turned off, 'Bar color' as green, 'Width' as 5, 'Border color' as white, and 'Border size' as 0 px. Below this is the 'Markers' section with 'Markers' turned on, 'Marker shape' as a square, and 'Size' as 6 px. At the bottom, 'Error band', 'Error labels', and 'Tooltip' are all turned off.

Section	Property	Value	
Error bars (Left Panel)	Apply settings to	Series: 07. Ref. line Position ...	
	Options	Enabled: On	
	Type	By field	
	Upper bound	11. Upper B...	
	Lower bound	+Add data	
	Relationship to measure	Relative	
	Make symmetrical	Off	
	Error bars (Right Panel)	Bar	On
		Match series color	Off
		Bar color	Green
Width		5	
Border color		White	
Border size		0 px	
Markers		On	
Marker shape		Square	
Size		6 px	
Error band		Off	
Error labels	Off		
Tooltip	Off		

## 2. Add Error Bar for Negative Relative Variance

- Select series 08. Ref. line Position neg. rel. var %
- Go to Options → Enabled.
- Set Lower Bound = [12. Lower Bound error bar neg. rel. var. %]
- Relation to measure = **Relative**.
- Turn Bars and Markers **On**
- Turn Error Bands, Error Label and Tooltip **Off**.
- Apply the remaining settings exactly as shown in the screenshots below.

The screenshot displays the 'Error bars' settings panel, which is organized into several sections. The 'Error bars' section at the top includes an 'Apply settings to' dropdown set to '08. Ref. line Position ...'. Below this is the 'Options' section, where 'Enabled' is turned on, 'Type' is set to 'By field', and the 'Upper bound' is set to '+Add data'. The 'Lower bound' is set to '12. Lower B...', and the 'Relationship to measure' is set to 'Relative'. The 'Make symmetrical' option is turned off. The 'Bar' section shows 'Match series color' is off, 'Bar color' is red, 'Width' is 5, 'Border color' is white, and 'Border size' is 0 px. The 'Markers' section shows 'Markers' is turned on, 'Marker shape' is a square, and 'Size' is 6 px. At the bottom, three options are turned off: 'Error band', 'Error labels', and 'Tooltip'.

▼ Error bars

Apply settings to

Series

08. Ref. line Position ... ▼

▼ Options

Enabled ☒

Type

By field ▼

Upper bound

+Add data

Lower bound

12. Lower B... ✕ | >

Relationship to measure

Relative ▼

Make symmetrical ☐

▼ Bar ☒

Match series color ☐

Bar color

Width

5

Border color

Border size

0 px

▼ Markers ☒

Marker shape

Size

6 px

> Error band ☐

> Error labels ☐

> Tooltip ☐

## Final Result

And you're done! You now have a fully functional **native Column Chart with Relative Variance**:


- Actual vs Comparison in columns
- Relative variance as scaled error bars
- Clean, controlled formatting with dynamic axis scaling
- Adaptable for any metric and comparison type

I hope you found this guide helpful!

Questions or feedback?

**Reach out:** Michel van Schaik

**LinkedIn:** <https://www.linkedin.com/in/michel-van-schaik/>



## Appendix – DAX Measures Cheat Sheet

#	Measure name	Type	Purpose / description	Visual placement
01	Month ≤ Selected Month	Helper	Returns 1 when the supplemental month ≤ selected month in main calendar; used to limit Actuals/labels to historical months.	Not placed (used by other measures)
02	Actual Value	Column	Actuals up to the selected month (via USERELATIONSHIP to supplemental calendar).	Column values
03	Comparison Value	Column	Comparison metric (e.g., PY) for the full year.	Column values
04	Data label Position abs. var.	Data Label position	Y-position anchor for the absolute variance label (max of Actual/Comparison), only for historical months.	<b>Line values</b> (lines turned OFF) (paired with #05 as Data Label Value)
05	Data label Value abs. var.	Data Label (value)	Absolute variance value (Actual vs Comparison), only for historical months.	<b>Data labels → Value field</b> (paired to #04)
06	Max Value of Columns	Helper	Returns the max of Actual vs Comparison across the selection; used to scale reference lines consistently.	Not placed (used by other measures)
07	Ref. line Position pos. rel. var %	Reference line	Baseline used for <b>positive</b> relative variance error bars.	<b>Line values</b> (lines turned OFF)
08	Ref. line Position neg. rel. var %	Reference line	Baseline used for <b>negative</b> relative variance error bars.	<b>Line values</b> (lines turned OFF)
09	Data label Value pos. rel. var. %	Data Label (value)	<b>Positive</b> relative variance (% with “+” sign via dynamic format string).	<b>Data labels → Value field</b> (paired to #13)
10	Data label Value neg. rel. var. %	Data Label (value)	<b>Negative</b> relative variance (%).	<b>Data labels → Value field</b> (paired to #14)
11	Upper Bound error bar pos. rel. var. %	Error bar (upper)	Upper bound for <b>positive</b> relative variance error bar. Scaled by 1.5 × and <b>capped at 50%</b> to avoid overlap.	<b>Attach as error bar upper bound</b> for the positive reference series (#07)
12	Lower Bound error bar neg. rel. var. %	Error bar (lower)	Lower bound for <b>negative</b> relative variance error bar. Scaled by 1.5 × and <b>capped at -50%</b> to avoid overlap.	<b>Attach as error bar lower bound</b> for the negative reference series (#08)
13	Data label Position pos. rel. var. %	Data Label (position)	Y-position anchor for <b>positive</b> relative % labels (built from #11 + #07).	<b>Line values</b> (lines turned OFF) (paired with #09 as Data Label Value)
14	Data label Position neg. rel. var. %	Data Label (position)	Y-position anchor for <b>negative</b> relative % labels (built from #12 + #08).	<b>Line values</b> (lines turned OFF) (paired with #10 as Data Label Value)
15	Max Y-axis Value	Max Y-Axis	It prevents data labels being cut off at chart top	Set as <b>Y-Axis → Maximum</b>